STAT436 **Homework 11** **Due Tuesday 3/25 at 2.30pm**

Directions: Students are to answer both exercises below showing all relevant work; conclusions and justifications are to be given in clear detailed English. Please type up your solutions or write very neatly.

1. Finney (1978, p.18) provides data regarding the tolerances of cats for tinctures of two variants of Strophanthus and Ouabain. The table below shows the fatal doses, or tolerances, with the data recorded as quantities per kg body weight; the last row of the table also provides the means. We wish to use these data to estimate the relative potency (a) Strophanthus1 to Strophanthus2, and (b) Strophanthus2 to Ouabain.

Strophanthus1	Strophanthus2	Ouabain
15.5	24.2	52.3
15.8	18.5	99.1
17.1	20.0	47.6
14.4	22.7	65.1
12.4	17.0	66.8
18.9	14.7	57.6
23.4	22.0	49.3
		45.8
		66.9
Mean = 16.8	Mean = 19.9	Mean = 61.2

   (a) Using SAS Program/Output A, make the first comparison (Strophanthus1 versus Strophanthus2), by providing 90%, 95%, and 99% Wald and PLCIs for the relative potency: calculate the three Wald intervals exactly by using the appropriate tabled critical values; for the likelihood intervals (PLCIs), just use your eyeball estimates. Comment on the necessary assumption here regarding the respective treatment variances and the reasonableness of this assumption for these data. Also, give the p-value for just accepting equivalent potency for the two drugs for both the Wald and PL situations.
   (b) Using SAS Program/Output B, answer each of the questions asked in part (a) but here for the second comparison: that is for comparing Strophanthus2 with Ouabain.

2. Return to the data discussed in Homework 8/Exercise 2. Our goal here is to assess the relative potency of dichlorodiphenyltrichloroethane (DDT) at 2.0% w/v to g-benzene hexachloride (gBHC) used at 1.5% w/v; SAS Program/Output C can be used here. The program runs PROC NLMIXED four times followed by some matrix code in SAS/IML and a graph.
   (a) Use the first or the second NLMIXED to give parameter estimates of the slope (β, called "bet" in the program) and the LD50 (γ, called “gam” in the program) for each of the curves (one for DDT and one for gBHC).
   (b) In contrasting the output of the second and third NLMIXEDs, perform a likelihood-based test of parallelism. Clearly write out your hypotheses, degrees of freedom, conclusion and p-value.
   (c) Assuming parallelism of the two curves, provide 90%, 95% and 99% Wald and PLCIs for the relative potency. Do you believe the treatments are equally potent? Clearly support your claim(s).

Homework 11 Appendix

SAS Program A (NLIN)

```
data one;
  do trt='stroph1','stroph2','ouabain';
    ni=7; if trt='ouabain' then ni=9;
  do rep=1 to ni;
    dum1=(trt='stroph1'); dum2=(trt='stroph2');
    input tols @@; tols=tols/10; logtols=log(tols); drop rep; output;
  end; end; datalines;
155 158 171 144 124 189 234 242 185 200 227 170 147 220
523 991 476 651 668 576 493 458 669
;
******  only comparing Strophanthus 1 with Strophanthus 2 now ******;
data one; set one; if trt='ouabain' then delete;
proc nlin hougaard data=one;
  parms th1=1 th2=1;
  model tols=th1*dum1+th1*th2*(1-dum1);
  output out=two r=r p=p;
run;
```

SAS Output A (NLIN)

The NLIN Procedure
Dependent Variable tols
Method: Gauss-Newton
Iterative Phase

Iter	th1	th2	Sum of Squares
0	1.0000	1.0000	4381.3
1	8.8929	2.5429	632.7
2	12.7285	1.2962	338.8
3	16.3728	1.1631	150.0
4	16.7816	1.1841	144.0
5	16.7857	1.1838	144.0

NOTE: Convergence criterion met.

Estimation Summary

Method	Gauss-Newton
Iterations	5
Observations Read	14
Observations Used	14
Observations Missing	0

Source	DF	Sum of Squares	Mean Square	F Value	Approx Pr > F
Model	1	33.3257	33.3257	2.78	0.1215
Error	12	144.0	12.0019		
Corrected Total	13	177.3			

Parameter	Estimate	Approx Std Error	Approximate 95% Confidence Limits		Skewness
th1	16.7857	1.3094	13.9327	19.6387	9.47E-17
th2	1.1838	0.1209	0.9204	1.4472	0.3576

Approximate Correlation Matrix

	th1	th2

```
                    th1         1.0000000       -0.7639278
                    th2        -0.7639278        1.0000000
```

Output A Residual Plot



SAS Program A (IML) – Used to find Profile Likelihood Graph (below)

```
proc iml;
  start neg2ll(th) global(yy,zz,n1,n2,nn);
    th1=th[1]; th2=th[2]; sig=th[3]; sig2=sig*sig;
    ym=yy-th1*j(n1,1); zm=zz-th1*th2*j(n2,1);
    brac=t(ym)*ym+t(zm)*zm;
    thingy=nn*log(sig2)+brac/sig2;
    return(thingy);
  finish neg2ll;

  n1=7; n2=7; nn=n1+n2;
  yy={15.5,15.8,17.1,14.4,12.4,18.9,23.4};
  zz={24.2,18.5,20.0,22.7,17.0,14.7,22.0};
  th0={16 1 3.5}; opt={.,0};
  con={.001 .001 .001,
         .     .     . };
  call nlptr(rc,theta,"neg2ll",th0,opt,con);
  minneg2ll=neg2ll(theta); print theta minneg2ll; ans=j(50,5);
  do jj=1 to 50;
    ans[jj,3]=minneg2ll+2.71;
    ans[jj,4]=minneg2ll+3.84;
    ans[jj,5]=minneg2ll+6.63;
    star=0.8; fini=1.6; th2=star+(fini-star)*(jj-1)/49; ans[jj,1]=th2;
    th0=theta; opt={.,0};
    con2={.001 .001 .001,
            .     .     . };
    th0[2]=th2; con2[1,2]=th2; con2[2,2]=th2;
    call nlptr(rc,thtil,"neg2ll",th0,opt,con2);
    ans[jj,2]=neg2ll(thtil);
  end;
  create dset from ans[colname={theta2 neg2ll cut90 cut95 cut99}];
  append from ans;
quit;
data final; set dset;
  y=neg2ll;  zz=1; output;
```

```
  y=cut90;  zz=2; output;
  y=cut95;  zz=3; output;
  y=cut99;  zz=4; output;

axis1 w=2 minor=none label=(a=90 font=swiss 'Profile Likelihood');
axis2 w=2 minor=none label=(font=swiss 'relative potency');
symbol1 i=join w=5 l=1 c=blue;
symbol2 i=join w=3 l=2 c=blue;
symbol3 i=join w=3 l=3 c=blue;
symbol4 i=join w=3 l=4 c=blue;
proc gplot data=final;
  plot y*theta2=zz / fr vaxis=axis1 nolegend haxis=axis2;
run;
```

SAS Program A (Output from above IML)

THETA			MINNEG2LL
16.785714	1.1838298	3.2073894	46.632806

SAS Program A (Profile Likelihood Plot)



SAS Program B (NLIN)

```
data one;
  do trt='stroph1','stroph2','ouabain';
    ni=7; if trt='ouabain' then ni=9;
  do rep=1 to ni;
    dum1=(trt='stroph1'); dum2=(trt='stroph2');
    input tols @@; tols=tols/10; logtols=log(tols); drop rep; output;
  end; end; datalines;
155 158 171 144 124 189 234 242 185 200 227 170 147 220
523 991 476 651 668 576 493 458 669
```

```
;
*********  only comparing Strophanthus 2 with Ouabain now **********;
data one; set one; if trt='stroph1' then delete;
proc nlin hougaard data=one;
  parms th1=20 th2=1;
  model tols=th1*dum2+th1*th2*(1-dum2);
  output out=two r=r p=p;
run;
```

SAS Output B (NLIN)

The NLIN Procedure
Dependent Variable tols
Method: Gauss-Newton
Iterative Phase

Iter	th1	th2	Sum of Squares
0	20.0000	1.0000	17491.9
1	19.8714	3.0648	2240.1
2	19.8714	3.0781	2239.5

NOTE: Convergence criterion met.

Source	DF	Sum of Squares	Mean Square	F Value	Approx Pr > F
Model	1	6714.6	6714.6	41.98	<.0001
Error	14	2239.5	160.0		
Corrected Total	15	8954.1			

Parameter	Estimate	Approx Std Error	Approximate 95% Confidence Limits		Skewness
th1	19.8714	4.7804	9.6185	30.1243	-157E-18
th2	3.0781	0.7703	1.4260	4.7302	1.3876

Approximate Correlation Matrix

	th1	th2
th1	1.0000000	-0.9613212
th2	-0.9613212	1.0000000

Output B Residual Plot



SAS Program B (NLMIXED)

```
proc nlmixed data=one;
  parms th1=20 th2=1 sig=15;
  mean=th1*dum2+th1*th2*(1-dum2);
  var=sig*sig*(dum2+th2*th2*(1-dum2));
  model tols~normal(mean,var);
run;
```

SAS Output B (NLMIXED)

The NLMIXED Procedure

Specifications

Data Set	WORK.ONE
Dependent Variable	tols
Distribution for Dependent Variable	Normal
Optimization Technique	Dual Quasi-Newton
Integration Method	None

Dimensions

Observations Used	16
Observations Not Used	0
Total Observations	16
Parameters	3

Iteration History

Iter	Calls	NegLogLike	Diff	MaxGrad	Slope
1	2	72.5970223	24.30564	19.37577	-507.827
2	4	68.1053188	4.491703	4.574078	-2.32329
3	6	67.2351444	0.870174	0.79893	-0.68407
4	8	67.0991129	0.136031	1.016443	-0.07806
5	11	65.5912049	1.507908	4.613051	-0.07998
6	12	63.7545742	1.836631	1.578168	-1.82374
7	15	59.407569	4.347005	4.438401	-3.51229
8	18	58.381766	1.025803	5.212151	-24.7037
9	20	57.9457613	0.436005	0.808497	-2.25305
10	22	57.9193164	0.026445	0.681404	-0.06901
11	24	57.7956232	0.123693	0.978052	-0.08105
12	26	56.7993382	0.996285	1.249764	-0.18279
13	29	56.2484918	0.550846	1.810055	-0.91634
14	31	56.1550605	0.093431	0.813316	-0.21916
15	33	56.1393852	0.015675	0.019057	-0.03707
16	35	56.1387074	0.000678	0.016339	-0.0019
17	37	56.1386143	0.000093	0.001652	-0.00019
18	39	56.138614	2.673E-7	0.00003	-5.56E-7

NOTE: GCONV convergence criterion satisfied.

Fit Statistics

-2 Log Likelihood	112.3
AIC (smaller is better)	118.3
AICC (smaller is better)	120.3
BIC (smaller is better)	120.6

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
th1	19.4708	1.5365	16	12.67	<.0001	0.05	16.2135	22.7280
th2	3.1926	0.3318	16	9.62	<.0001	0.05	2.4891	3.8960
sig	4.2070	0.7833	16	5.37	<.0001	0.05	2.5465	5.8676

SAS Program B (IML) – Used to find Profile Likelihood Graph (below)

```
proc iml;
  start neg2ll(th) global(yy,zz,n1,n2,nn);
    th1=th[1]; th2=th[2]; sig=th[3]; sig2=sig*sig;
    ym=yy-th1*j(n1,1); zm=zz-th1*th2*j(n2,1);
    brac=th2*th2*t(ym)*ym+t(zm)*zm;
    thingy=nn*log(sig2)+2*n2*log(th2)+brac/(th2*th2*sig2);
    return(thingy);
  finish;

  n1=7; n2=9; nn=n1+n2;
  yy={24.2,18.5,20.0,22.7,17.0,14.7,22.0};
  zz={52.3,99.1,47.6,65.1,66.8,57.6,49.3,45.8,66.9};
  th0={16 3 12}; opt={.,0};
  con={.001 .001 .001,
          .    .    . };
  call nlptr(rc,theta,"neg2ll",th0,opt,con);
  minn2ll=neg2ll(theta); print theta minn2ll; ans=j(50,5);
  do jj=1 to 50;
    ans[jj,3]=minn2ll+2.71;
    ans[jj,4]=minn2ll+3.84;
    ans[jj,5]=minn2ll+6.63;
    star=2.2; fini=4.5; th2=star+(fini-star)*(jj-1)/49; ans[jj,1]=th2;
    th0=theta; opt={.,0};
    con2={.001 .001 .001,
            .    .    . };
    th0[2]=th2; con2[1,2]=th2; con2[2,2]=th2;
    call nlptr(rc,thtil,"neg2ll",th0,opt,con2);
    ans[jj,2]=neg2ll(thtil);
  end;
  create dset from ans[colname={theta2 neg2ll cut90 cut95 cut99}];
  append from ans;
quit;
data final; set dset;
  y=neg2ll;  zz=1; output;
  y=cut90;  zz=2; output;
  y=cut95;  zz=3; output;
  y=cut99;  zz=4; output;

axis1 w=2 minor=none label=(a=90 font=swiss 'Profile Likelihood');
axis2 w=2 minor=none label=(font=swiss 'relative potency');
symbol1 i=join w=5 l=1 c=blue;
symbol2 i=join w=3 l=2 c=blue;
symbol3 i=join w=3 l=3 c=blue;
symbol4 i=join w=3 l=4 c=blue;
proc gplot data=final;
  plot y*theta2=zz / fr vaxis=axis1 nolegend haxis=axis2;
run;
```

SAS Program B (Output from above IML)

THETA	MINN2LL
19.47071 3.1925748 4.2070397	82.871195

SAS Program B (Profile Likelihood Plot)



SAS Program C (First NLMIXED)

```
data insect;
  do type='DDT ','gBHC';
  do deposit=2,2.64,3.48,4.59,6.06,8;
    ldep=log(deposit); dd=(type='DDT '); dg=(type='gBHC');
    input y n @@; output;
  end; end; datalines;
3 50 5 49 19 47 19 38 24 49 35 50
2 50 14 49 20 50 27 50 41 50 40 50
;
proc sort; by type; run;
proc nlmixed;
  parms bet=2.5 gam=1;
  eta=bet*(ldep-log(gam));
  expeta=exp(eta);
  p=expeta/(1+expeta);
  model y~binomial(n,p);
  by type;
run;
```

SAS Output C (First NLMIXED)

The NLMIXED Procedure
Specifications

Data Set	WORK.INSECT
Dependent Variable	y
Distribution for Dependent Variable	Binomial
Optimization Technique	Dual Quasi-Newton
Integration Method	None

---------- type=DDT ----------

The NLMIXED Procedure

Dimensions

Observations Used	6
Observations Not Used	0
Total Observations	6
Parameters	2

Parameters

bet	gam	NegLogLike
2.5	1	416.03576

Iteration History

Iter	Calls	NegLogLike	Diff	MaxGrad	Slope
1	3	102.564578	313.4712	82.7474	-2532.41
2	5	47.8553472	54.70923	21.29345	-36.046
3	7	37.4773597	10.37799	15.25714	-5.69886
4	9	31.3603422	6.117017	17.30058	-6.46413
5	10	20.8812123	10.47913	4.740603	-31.1141
6	12	18.6311778	2.250035	4.517671	-4.1283
7	15	17.9103999	0.720778	7.213801	-0.9354
8	16	16.8778268	1.032573	3.104732	-0.73744
9	17	16.4674615	0.410365	2.628793	-1.05518
10	18	16.0158884	0.451573	0.303515	-1.30892
11	20	16.0096329	0.006255	0.024832	-0.01262
12	22	16.0095708	0.000062	0.001178	-0.0001
13	24	16.0095707	7.964E-8	3.772E-6	-1.58E-7

NOTE: GCONV convergence criterion satisfied.

Fit Statistics

-2 Log Likelihood	32.0
AIC (smaller is better)	36.0
AICC (smaller is better)	40.0
BIC (smaller is better)	35.6

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr > |t|	Alpha	Lower	Upper
bet	2.2824	0.3190	6	7.16	0.0004	0.05	1.5019	3.0629
gam	5.3569	0.3505	6	15.28	<.0001	0.05	4.4993	6.2145

---------- type=gBHC ----------

The NLMIXED Procedure

Dimensions

Observations Used	6
Observations Not Used	0
Total Observations	6
Parameters	2

Parameters

bet	gam	NegLogLike
2.5	1	315.327115

Iteration History

Iter	Calls	NegLogLike	Diff	MaxGrad	Slope
1	3	69.6343928	245.6927	67.9755	-2118.97
2	5	32.6585765	36.97582	18.38442	-23.0892
3	7	19.7707462	12.88783	9.070433	-6.82388

```
                  4         9    17.8804383    1.890308    6.436566    -1.08876
                  5        11    15.8282969    2.052141     2.38946    -2.15176
                  6        13    15.7019341    0.126363    0.433908    -0.29391
                  7        15    15.6881118    0.013822     0.02003    -0.02471
                  8        17     15.688085    0.000027    0.000488    -0.00005
                  9        19     15.688085    5.247E-9    9.171E-6    -1.07E-8
                        NOTE: GCONV convergence criterion satisfied.

                                      Fit Statistics
                                -2 Log Likelihood                  31.4
                                AIC (smaller is better)            35.4
                                AICC (smaller is better)           39.4
                                BIC (smaller is better)            35.0

                                   Parameter Estimates
                             Standard
Parameter   Estimate        Error    DF  t Value  Pr > |t|   Alpha     Lower     Upper
bet           2.8381       0.3392     6     8.37    0.0002    0.05    2.0080    3.6681
gam           4.1558       0.2021     6    20.57    <.0001    0.05    3.6613    4.6502
```

SAS Program C (Second NLMIXED)

```
proc nlmixed;
  parms betd=2 betg=3 gamd=5 gamg=4;
  bet=betd*dd+betg*dg; gam=gamd*dd+gamg*dg;
  t=(deposit/gam)**bet;
  p=t/(1+t);
  model y~binomial(n,p);
run;
```

SAS Output C (Second NLMIXED)

```
                                The NLMIXED Procedure
                                    Specifications
                        Data Set                                    WORK.INSECT
                        Dependent Variable                          y
                        Distribution for Dependent Variable         Binomial
                        Optimization Technique                      Dual Quasi-Newton
                        Integration Method                          None

                                         Dimensions
                                  Observations Used                      12
                                  Observations Not Used                   0
                                  Total Observations                     12
                                  Parameters                              4

                                         Parameters
                               betd        betg        gamd        gamg    NegLogLike
                                  2           3           5           4    33.5095283

                                      Iteration History
                        Iter     Calls    NegLogLike         Diff      MaxGrad       Slope
                           1         3     31.920406     1.589122     2.385246    -65.4072
                           2         5    31.7104028     0.210003     0.406192    -9.35406
                           3         7    31.7000375     0.010365     0.257504    -0.19118
                           4         9    31.6991427     0.000895     0.204831    -0.00847
```

```
                5        10    31.6977469    0.001396    0.051084    -0.00542
                6        12    31.6976557    0.000091    0.000121    -0.00018
                7        14    31.6976557     1.58E-9    2.016E-6    -3.18E-9

                      NOTE: GCONV convergence criterion satisfied.

                                  Fit Statistics
                          -2 Log Likelihood                  63.4
                          AIC (smaller is better)            71.4
                          AICC (smaller is better)           77.1
                          BIC (smaller is better)            73.3

                                Parameter Estimates
                        Standard
Parameter   Estimate      Error    DF  t Value  Pr > |t|   Alpha     Lower     Upper
betd          2.2824     0.3190    12     7.16    <.0001    0.05    1.5874    2.9774
betg          2.8381     0.3392    12     8.37    <.0001    0.05    2.0990    3.5772
gamd          5.3569     0.3505    12    15.28    <.0001    0.05    4.5932    6.1206
gamg          4.1558     0.2021    12    20.57    <.0001    0.05    3.7155    4.5961
```

SAS Program C (Third NLMIXED)

```
proc nlmixed;
  parms bet=2.5 gamd=5 gamg=4;
  gam=gamd*dd+gamg*dg;
  t=(deposit/gam)**bet;
  p=t/(1+t);
  model y~binomial(n,p);
run;
```

SAS Output C (Third NLMIXED)

```
                          The NLMIXED Procedure
                              Specifications
              Data Set                                    WORK.INSECT
              Dependent Variable                          y
              Distribution for Dependent Variable         Binomial
              Optimization Technique                      Dual Quasi-Newton
              Integration Method                          None

                                Dimensions
                        Observations Used                      12
                        Observations Not Used                   0
                        Total Observations                     12
                        Parameters                              3
                                Parameters
                          bet        gamd        gamg    NegLogLike
                          2.5           5           4    33.1943149

                              Iteration History
              Iter     Calls    NegLogLike        Diff     MaxGrad       Slope
                 1         3    32.4611226    0.733192    0.859125    -38.4815
                 2         6    32.4129924     0.04813    0.144567    -1.78525
                 3         9    32.4123667    0.000626     0.06799    -0.03425
                 4        10    32.4121859    0.000181     0.00052    -0.00036
                 5        11    32.4121859    2.497E-8    0.000026    -4.81E-8
```

```
NOTE: GCONV convergence criterion satisfied.

                    Fit Statistics
                -2 Log Likelihood                  64.8
                AIC (smaller is better)            70.8
                AICC (smaller is better)           73.8
                BIC (smaller is better)            72.3

                    Parameter Estimates
                          Standard
Parameter   Estimate        Error    DF  t Value  Pr > |t|  Alpha    Lower    Upper
bet           2.5576       0.2322    12    11.02    <.0001   0.05   2.0518   3.0635
gamd          5.2621       0.3009    12    17.49    <.0001   0.05   4.6065   5.9176
gamg          4.1635       0.2187    12    19.04    <.0001   0.05   3.6871   4.6399
```

SAS Program C (Fourth NLMIXED)

```
proc nlmixed;
  parms bet=2.5 gamd=5 rho=2;
  gamg=rho*gamd; gam=gamd*dd+gamg*dg;
  t=(deposit/gam)**bet;
  p=t/(1+t);
  model y~binomial(n,p);
run;
```

SAS Output C (Fourth NLMIXED)

```
                    The NLMIXED Procedure
                        Specifications
Data Set                                    WORK.INSECT
Dependent Variable                          y
Distribution for Dependent Variable         Binomial
Optimization Technique                      Dual Quasi-Newton
Integration Method                          None

                        Dimensions
                Observations Used                  12
                Observations Not Used               0
                Total Observations                 12
                Parameters                          3

                        Parameters
                  bet        gamd         rho    NegLogLike
                  2.5           5           2    155.448559

                    Iteration History
Iter     Calls    NegLogLike        Diff     MaxGrad       Slope
   1         3    51.1032767    104.3453    77.76662    -938.654
   2         5    47.1353752    3.967902    22.18537    -53.4762
   3         7    33.5502428    13.58513    22.58646    -13.2707
   4         8    33.2342302    0.316013    28.80981    -2.67963
   5        10     32.509376    0.724854    1.631572    -1.60408
   6        12    32.4137701    0.095606    0.546839    -0.10035
   7        14    32.4121872    0.001583    0.012647    -0.00301
   8        16    32.4121859    1.328E-6    0.001436    -2.52E-6
   9        18    32.4121859    2.963E-9    0.000052    -5.42E-9
```

```
NOTE: GCONV convergence criterion satisfied.

                    Fit Statistics
-2 Log Likelihood                  64.8
AIC (smaller is better)            70.8
AICC (smaller is better)           73.8
BIC (smaller is better)            72.3

                    Parameter Estimates
                  Standard
Parameter  Estimate     Error    DF  t Value  Pr > |t|  Alpha    Lower    Upper
bet          2.5576    0.2322    12    11.02    <.0001   0.05   2.0518   3.0635
gamd         5.2621    0.3009    12    17.49    <.0001   0.05   4.6065   5.9176
rho          0.7912   0.06119    12    12.93    <.0001   0.05   0.6579   0.9246
```

SAS Output C (IML Program to create output and Graph)

```
proc iml;
  start neg2ll(th) global(depd,depg,nd,ng,yd,yg);
     bet=th[1]; gamd=th[2]; rho=th[3]; gamg=rho*gamd;
     td=(depd/gamd)##bet; tg=(depg/gamg)##bet; pd=td/(1+td); pg=tg/(1+tg);
     tomax=t(yd)*log(td)-t(nd)*log(1+td)+t(yg)*log(tg)-t(ng)*log(1+tg);
     return(-2*tomax);
  finish;

  depd={2,2.64,3.48,4.59,6.06,8}; depg={2,2.64,3.48,4.59,6.06,8};
  nd={50,49,47,38,49,50}; ng={50,49,50,50,50,50};
  yd={3,5,19,19,24,35}; yg={2,14,20,27,41,40};
  th0={2.5 5 .5}; opt={.,0};
  con={.001 .001 .001,
         .     .     . };
  call nlptr(rc,theta,"neg2ll",th0,opt,con);
  minn2ll=neg2ll(theta); print theta minn2ll; ans=j(50,5);
  do jj=1 to 50;
    ans[jj,3]=minn2ll+2.71;
    ans[jj,4]=minn2ll+3.84;
    ans[jj,5]=minn2ll+6.63;

    star=0.6; fini=1; rho=star+(fini-star)*(jj-1)/49; ans[jj,1]=rho;
    th0=theta; opt={.,0};
    con2={0.001 0.001 0.001,
            .     .     . };
    th0[3]=rho; con2[1,3]=rho; con2[2,3]=rho;
    call nlptr(rc,thtil,"neg2ll",th0,opt,con2);
    ans[jj,2]=neg2ll(thtil);
  end;
  create dset from ans[colname={rho negll cut90 cut95 cut99}];
  append from ans;
quit;
data final; set dset;
  y=negll;  zz=1; output;
  y=cut90;  zz=2; output;
  y=cut95;  zz=3; output;
  y=cut99;  zz=4; output;

axis1 w=2 minor=none label=(a=90 font=swiss 'Profile Likelihood');
axis2 w=2 minor=none label=(font=swiss 'relative potency');
symbol1 i=join w=5 l=1 c=blue;
symbol2 i=join w=3 l=2 c=blue;
```

```
symbol3 i=join w=3 l=3 c=blue;
symbol4 i=join w=3 l=4 c=blue;
title 'Profile Likelihood Graph';
proc gplot data=final;
  plot y*rho=zz / fr vaxis=axis1 nolegend haxis=axis2;
run;
```

THETA	MINN2LL
2.55762 5.2620783 0.7912255	627.35732